

Les fichiers SCRIPT

AutoCAD
Versions 2000 - 2000i
2002 - 2004
2005 - 2006

Dans la personnalisation d'AutoCAD il y a :

- la programmation par l'intermédiaire des langages ARX, VBA AutoLISP,
- les menus déroulants, les barres d'outils, le menu écran, voire même la table à digitaliser,
- les raccourcis clavier,
- les fichiers SCRIPT.

Ces derniers permettent d'automatiser des tâches répétitives sans une connaissance de la programmation, simplement en mettant bout-à-bout des commandes AutoCAD telles que vous les taperiez au clavier. On appelle cela des macro-commandes. Elles seront exécutées par AutoCAD comme une suite de commandes. Voyons comment créer et exécuter un fichier SCRIPT.

L'écriture d'un fichier SCRIPT

Tout d'abord un fichier SCRIPT est un fichier de type ASCII, aussi appelé aussi Texte, qui peut être ouvert et modifié depuis le Bloc-Notes ou NotePad de Windows.

Son contenu correspond à des commandes ou à des options de commande AutoCAD. Par exemple pour exécuter la commande «Zoom» «étendu» on écrira :

Zoom Et

ou

**Zoom
Et**


Les commandes peuvent être écrites en anglais pour pouvoir être reconnues quelle que soit la langue d'AutoCAD.

Pour écrire un fichier SCRIPT il est nécessaire de connaître les commandes à utiliser ainsi que leurs options. Pas besoin de les apprendre par coeur, car il existe une méthode simple et pratique qui permettra de trouver le nom de la commande et ses différentes options.

Pour cela, il vous suffit de vous placer, bien évidemment sur AutoCAD, de lancer la commande, soit depuis le menu déroulant, soit depuis une icône, et de taper manuellement ses différentes options selon la tâche que vous souhaitez réaliser. Une fois le traitement de la commande terminé, tapez sur la touche « F2 » pour agrandir la fenêtre texte

AutoCAD afin de visualiser toute la syntaxe de votre tâche. Placez-vous ensuite dans le fichier SCRIPT et retranscrivez toutes les actions.

Prenons un exemple concret. Soit à automatiser la création de 4 cercles d'un diamètre 20, distants de 50 et placé 2 par deux. La coordonnée du centre du premier cercle sera de 0,0.



1. Lancez la commande « **Cercle** » ou « **_Cercle** », depuis l'icône de la barre d'outils dessin.
2. Spécifiez le centre en tapant au clavier « **0,0** », puis validez.
3. Tapez l'option « **D** » pour diamètre, puis validez.
4. Entrez la valeur du diamètre, soit **20**, puis validez.

Le premier cercle étant tracé, maintenant passons aux autres cercles.

5. Lancez de nouveau la commande « **Cercle** » ou « **_Cercle** ».
6. Spécifiez le centre en tapant au clavier « **50,0** », puis validez.
7. Tapez l'option « **D** » pour diamètre, puis validez.
8. Entrez la valeur du diamètre, soit **20**, puis validez.
9. Lancez de nouveau la commande « **Cercle** » ou « **«** ».
10. Spécifiez le centre en tapant au clavier « **50,50** », puis validez.

11. Tapez l'option « **D** » pour diamètre, puis validez.
12. Entrez la valeur du diamètre, soit **20**, puis validez.
13. Lancez encore une fois la commande « Cercle » ou « **_Cercle** ».
14. Spécifiez le centre en tapant au clavier « **0,50** », puis validez.
15. Tapez l'option « **D** » pour diamètre, puis validez.
16. Entrez la valeur du diamètre, soit **20**, puis validez.

Maintenant, ouvrez le Bloc-Notes de Windows, puis tapez les lignes suivantes en commençant à la première ligne et sans mettre un espace devant le premier caractère :

```
; Mon premier SCRIPT (ceci est une ligne de commentaire car elle commence par un point-virgule).  
; Création de 4 cercles d'un diamètre 20, distants de 50, disposés 2 par 2.  
Cercle  
0,0  
D  
20  
Cercle  
50,0  
D  
  
Cercle  
50,50  
D  
  
Cercle  
0,50  
D  
  
; fin du script
```

Notez la syntaxe, qui se résume à la commande, les options de la commande, les lignes de commentaires, qui commencent par le signe point-virgule, et des lignes vides.

Il est important de savoir que, dans un fichier SCRIPT, les lignes vides ou les espaces sont l'équivalent de la touche « **Entrée** ».

Avec moins de lisibilité, ce fichier SCRIPT aurait pu s'écrire sur une même ligne comme ceci :

```
Cercle 0,0 D 20 Cercle 50,0 D Cercle 50,50 D Cercle 0,50 D
```

Vous remarquerez qu'il est difficile de savoir qu'entre le « **D** », équivalent à l'option diamètre, et la prochaine commande « **Cercle** » il y a deux espaces. Encore moins que la fin de la ligne se termine par deux espaces. Donc attention à bien placer les caractères « *Espace* » au risque

de stopper le déroulement du SCRIPT suite à une mauvaise interprétation de la commande par AutoCAD ou à doubler la commande.

Sauvegardez le fichier sous un nom en lui attribuant l'extension de fichier « **SCR** » et non pas « **TXT** », par exemple « *MonScript.SCR* ».

Exécuter un fichier SCRIPT

L'exécution d'un fichier SCRIPT peut être réalisée à deux niveaux :

- En tapant la commande « **SCRIPT** » dans AutoCAD.
- Au lancement d'AutoCAD.

Depuis AutoCAD

1. Dans AutoCAD tapez la commande « **SCRIPT** ».
2. Sélectionnez le fichier SCRIPT, par exemple celui que vous venez de créer.

Si vous n'avez fait aucune faute de syntaxe, le déroulement du script doit arriver à sa fin sans problème, sinon il s'arrêtera. Vous devrez alors voir pourquoi le SCRIPT s'est arrêté, corriger le fichier et relancer le fichier script à nouveau.

Au lancement d'AutoCAD


Quelquefois il est intéressant que le fichier SCRIPT s'exécute à l'ouverture d'AutoCAD. Dans ce cas il est nécessaire de lancer AutoCAD depuis la commande « **Exécuter** » de Windows ou depuis son icône de lancement, en modifiant quelque peu sa syntaxe de lancement depuis l'option des propriétés.

Dans tous les cas, la syntaxe de lancement sera la suivante :

Nom_du_fichier_à_lancer /B Nom_du_fichier_Script
par exemple :

```
«C:\Program files\AutoCAD 2006\acad.exe /B C:\Dossier_Script\MonScript.scr»
```

Notez bien la syntaxe « **/B** » qui est obligatoire et qui donne ordre à AutoCAD d'exécuter le nom du fichier script qui suit.



Possibilités

- Un fichier SCRIPT peut contenir, en plus des commandes de bases, des commandes provenant d'applications telles que ARX, VBA ou AutoLISP. Dans ce cas il est nécessaire de connaître le nom du fichier et le nom de la commande. Dans le cas où le fichier ne serait pas chargé, il faudra au préalable le charger, par exemple :

```
; Chargement du fichier AutoLISP «MonFichierLisp.LSP»  
(load «MonFichierLisp»  
; Exécution de la commande provenant de ce fichier  
MaCommandeLisp
```

```
; Chargement du fichier ARX «MonFichierArx.ARX»  
(Arxload «MonFichierArx»  
; Exécution de la commande provenant de ce fichier  
MaCommandeArx
```

```
; Chargement et exécution du fichier VBA «MonFichierVBA.DVB»  
(VL-VBARUN «MonFichierVBA!Module.Commande»)
```

- Un Script peut être lancé en boucle si la dernière commande du fichier script est : **RSCRIPT**.
- Un SCRIPT peut être arrêté en cours de traitement, suite à une erreur de syntaxe ou par l'action de la touche **ESC**. Pour le relancer là où il s'est arrêté tapez la commande : **REPRENDRE** ou **_RESUME**.
- Les noms des commandes et des options peuvent être écrits aussi bien en français qu'en anglais.
- Un SCRIPT est considéré comme une commande qui peut être annulée par une seule commande **U**.

Limitations

- Un fichier SCRIPT ne permet pas une pause utilisateur, c'est-à-dire qu'en cours d'exécution il ne vous sera pas

possible de stopper le traitement pour entrer une valeur ou pour sélectionner des objets.

- Un SCRIPT peut lancer un autre SCRIPT. Dans ce cas le dernier SCRIPT prendra la main et le SCRIPT précédent deviendra inopérant.
- Si une information contient des espaces elle doit être placée entre des guillemets. Par exemple un nom de fichier long sera écrit de cette manière : « **C:\Les Cahiers d'AutoCAD\MonBloc** » et non pas *C:\Les Cahiers d'AutoCAD\MonBloc*.
- L'ouverture de certaines boîtes de dialogue arrête de traitement du fichier SCRIPT. Dans ce cas il faudra faire précéder le nom de la commande par le signe moins «-» pour obtenir les instructions en tant que ligne de commande.
- Un fichier SCRIPT ne peut pas contenir des fonctions DIESEL (hélas !).

Différence entre un fichier SCRIPT et une macro d'un BOUTON

Il y a quelques différences entre une macro provenant d'un fichier SCRIPT et une macro provenant d'un Bouton :

- La macro d'un bouton peut contenir des pauses utilisateur.
- La macro d'un SCRIPT ne peut pas contenir des fonctions DIESEL.

Idées de SCRIPT

- Insertion d'un format papier sur un espace de présentation,
- Impression de plusieurs documents DWG au lancement d'AutoCAD,
- Impression de différents onglets de même nom,
- Mise à jour de différentes variables systèmes,
- Etc.

Routines issues des cahiers **AutoCAD – Editions Dominique VAQUAND**

INFORMATIQUE CAO-DAO – Votre partenaire Dominique VAQUAND Sarl

**Développements sur AutoCAD – Formations tous niveaux sur site
Prestations de service – Vente et mise à jour AutoCAD**

BP 33 – 13430 EYGUIERES Cedex – Tél. : 04 90 57 96 70 – Fax : 04 90 57 96 23

E-mail : dominique.vaquand@free.fr – [Http://dominique.vaquand.free.fr](http://dominique.vaquand.free.fr)